# Maintainer: @naveen521kk on Github, Naveen M K <naveen@syrusdark.website>

_perlname=Win32-WinError
_realname="${_perlname,,}"
pkgbase=mingw-w64-perl-${_realname}
pkgname=("${MINGW_PACKAGE_PREFIX}-perl-${_realname}")
pkgver=0.04
pkgrel=2
pkgdesc="Perl module defining Windows error constants (mingw-w64)"
arch=('any')
url="https://metacpan.org/release/Win32-WinError"
groups=("${MINGW_PACKAGE_PREFIX}-perl-modules")
depends=(
  "${MINGW_PACKAGE_PREFIX}-perl"
)
options=('!emptydirs')
makedepends=("${MINGW_PACKAGE_PREFIX}-python")
license=('perl_5')
source=(
  "https://cpan.metacpan.org/authors/id/J/JD/JDB/${_perlname}-${pkgver}.tar.gz"
  "patchmakefile.py"
)
sha512sums=('6fa0b066ae284dd51b0461426903e2d323c959a92d4a65ce441a0d351d18ce839b536de054e7abd924f443a0e132f0ce8206f50c3fbe4a4d09c23bd147e95610'
            'd7ec5ba4a3d75f674f4028e8a53bc4177ab71a379bc63affebf910e1d5a0e491b6642318218b55ddc4d90e0125d61e4b97626782af57d44522bcb5543fb21cb2')

build() {
  cd "$srcdir/${_perlname}-${pkgver}"
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL
  export pkgdir=$pkgdir
  python $srcdir/patchmakefile.py
  mingw32-make
}

package() {
  cd "$srcdir/${_perlname}-${pkgver}"
  mingw32-make DESTDIR="${pkgdir}/" install
}
